

## 第4章 モナド

モナド (monad) は、Haskell (あるいは他の関数型言語) で破壊的代入 (変数の値など状態を変更すること) や入出力のような、他の言語では“副作用” (side effect) として実現される特徴を扱うための手法である。

もともとは数学のカテゴリー理論 (圏論) で使われている言葉を借用したものであるが、Haskell で使用するときには、背景となるカテゴリー理論のことを知っている必要はない。

### 4.1 参照透明性と副作用

純粋な関数型言語には、式は値を表すためだけのものであり、「変数の出現はその定義の右辺の式で置き換えても全体の意味は変わらない」という性質がある。このような性質を 参照透明性 (referential transparency) と呼び、プログラムの“意味”を考察していく上でとても重要な性質である。(参照透明性のおかげで、帰納法などによる証明が容易になる。) 一方、副作用は、式がその 副作用を持つ のことである。式が副作用を持ちうるということは、その言語では参照透明性が成り立たない、ということである。Haskell の式は副作用を持っていない。

C のような言語では、入出力や破壊的代入を扱う部分では、副作用を使っているため参照透明性は成り立たない。例えば、C で、

```
c = getchar(); putchar(c); putchar(c); // C-code ①
```

と

```
putchar(getchar()); putchar(getchar()); // C-code ②
```

とは、getchar() が副作用を持っているために異なるプログラムである。(上のプログラムでは1文字、下のプログラムでは2文字の入力が消費される。)

一見、参照透明性と入出力や破壊的代入は相容れない性質のように見える。しかし、Haskell では次のように考える。

入出力や、変数などの状態の変更は、何らかの“アクション”であり、単なる値とは異なる型を持っている。副作用を持つC言語などの関数に対応する Haskell の関数は、このアクションを含めて戻り値として返す関数として考える。(つまりアクションを“副”作用ではなく、一人前の値として扱う。) この“アクション”の型は、アクションに対応している文脈でしか使用することができない。従って、アクションと値は区別され、参照透明性が保たれる。

例えば、Haskell で `getchar`, `putchar` に対応する関数は、それぞれ次のような型を持っている。

```
getChar :: _____  
putChar :: _____
```

この `IO` という型構成子がアクションの型を表す。そもそも、C 言語の `putchar(getchar ())` という式に対応する `putChar getChar` のような Haskell の式は、型エラーになってしまう。IO 型に用意されている演算子 `>>=` を用いて、次のように書かなければいけない。

```
getChar >>= (\ c -> putChar c)
```

ここで、`>>=` の型は `IO a -> (a -> IO b) -> IO b` である。(後述するように、実際にはより一般的な型を持っている。) この式では、`getChar` というアクションの結果得られる `Char` 型の値が、`c` という変数に束縛され、続いて `putChar c` というアクションが実行される。アクションの型を持つ式から値だけを抽出するには、`>>=` のような演算子を介する必要がある。`Char` 型の変数 `c` に対して、`c = getChar` と書けるわけではないので、型の面からも `c` が `getChar` と置き換えられないことが明白である。

**Q 4.1.1** C-code ①と C-code ②に対応する Haskell プログラム(の断片)を `getChar`, `putChar`, `>>=` を使って書け。

---

---

---

---

---

副作用を持つプログラムは実行する順番により意味が変わるので、並列に実行されるプログラムでは問題となることがある。参照透明性を保ちながらアクションを扱えることは理論的に有用であるだけでなく、並列コンピューティングなど実用面でも必要となってきた。

## 4.2 モナドとは

このような、さまざまな言語で副作用として実現される特徴(入出力・破壊的代入・例外処理・非決定性など)に対するアクションの型が、実は共通の構造を持っていて、同じ構造を持つ演算子で取り扱えることがわかっている。この共通の構造を持つ型(正確には型構成子)のことを \_\_\_\_\_ (monad) という。つまり、モナドはアクションの型である。上記の `IO` もモナドである。ただし、モナドの

中にはアクションという言葉がふさわしくないものもあるので、以下では代わりに“計算”(computation)という言葉を使う。

具体的にはモナドとは

$$\begin{aligned} \text{return} &:: a \rightarrow M a \\ (>>=) &:: M a \rightarrow (a \rightarrow M b) \rightarrow M b \end{aligned}$$

という型の関数の存在する型構成子  $M$  のことである。より厳密には、 $\text{return}$ ,  $(>>=)$  が

$$\begin{aligned} (\text{return } a) >>= k &= k a \\ m >>= (\lambda a \rightarrow \text{return } a) &= m \\ (m_1 >>= k_1) >>= k_2 &= m_1 >>= (\lambda a \rightarrow (k_1 a >>= k_2)) \end{aligned}$$

の3つの等式(monad law)を満たす必要がある。これらの等式は、直感的には  $\text{return}$  が  $(>>=)$  の“単位元”であること、 $(>>=)$  が結合則を満たすこと、を示している。つまり、数に対する、加算(+)と加算の単位元0の次の等式に相当する。

$$\begin{aligned} 0 + x &= x \\ x + 0 &= x \\ (x + y) + z &= x + (y + z) \end{aligned}$$

直観的には  $M a$  という型は、\_\_\_\_\_の型を意味する。また、 $\text{return}$  と  $(>>=)$  の直観的な意味は次のとおりである。

- $\text{return } a \dots$  値  $a$  をそのまま返す **何もしない(アクションが実質ない)** 計算を表す。
- $m >>= k \dots$  まず、 $m :: M a$  を評価し、その結果の値を関数  $k :: a \rightarrow M b$  に渡して、続けて評価する計算を表す。

### 4.3 モナドと型クラス

モナド  $M$  の定義は模倣したい副作用により異なるし、付随する関数  $\text{return}$ ,  $(>>=)$  の定義ももちろん異なる。しかし、 $\text{return}$ ,  $(>>=)$  は  $M$  をパラメータとして、決まった型を持つので、型クラスを用いてアドホック多相な関数として定義することができる。

```
1 class Monad m where
2   return :: a -> m a
3   (>>=)  :: m a -> (a -> m b) -> m b
```

実際は、Monad は型ではなく、型構成子に対する型クラス(正確には型構成子クラス)になっている。

## 4.4 IO モナド

IO は Haskell の Prelude (最初から読み込まれているライブラリ) に入っているモナドである。型クラス `Monad` のインスタンスである。その定義は一般のプログラマからは見えない組込みの型となっている。ただし、直観的には次のような定義を持つ型だと考えることができる。

```
1 -- type IO a ≡ _____
2
3 -- instance Monad IO where
4 -- -- 注: 実際には type で定義された型名を instance には使えない。
5 -- return a = \ w -> (a,w)
6 -- m >>= k = \ w -> let (a,w1) = m w in k a w1
```

ただし `RealWorld` は、コンピューター全体のファイルなどの状態を表す型である。引数の `RealWorld` 型のデータは隠されていて、他の計算で使われないことが保証できるため、破壊的に書き換えて、戻り値の `RealWorld` 型のデータのために使っても良い、ということである。

IO は Haskell で入出力や状態を効率的に扱うために、処理系で特別な扱いを受ける。`putChar`, `getChar` の他にも、主なプリミティブとして以下のような関数がある。

```
1 putStr      :: String -> IO () -- 文字列を出力する
2 putStrLn   :: String -> IO () -- 文字列を出力して、改行する
3
4 getLine    :: IO String      -- 一行を読み込む
5 getContents :: IO String     -- EOF まで読み込む
6
7 print     :: Show a => a -> IO () -- 文字列に変換して出力し、改行する
8 readLn   :: Read a => IO a      -- 一行を読み込んでパースする
```

これらの他にファイルに対して入出力するための関数なども用意されている。

`getContents` は遅延評価でファイルの内容を読み込む。つまり次のようなプログラムでは、ファイルのすべてを読み込むことはなく、最初の 1 行のみを出力する。

```
1 main :: IO ()
2 main = getContents >>=
3       (\ all -> let ls = lines all
4                 in putStrLn (head ls))
```

ただし、`lines :: String -> [String]` は文字列を改行文字で分割する関数である。

さらに、`Data.IORef` というモジュールを `import` することで、破壊的代入が可能な参照型 (`IORef`) を扱う関数も用意される。

```
1 -- import Data.IORef が必要
2
3 newIORef  :: a -> IO (IORef a) -- 新しい参照の作成
4 readIORef :: IORef a -> IO a  -- 参照の読出し
```

```
5 writeIORef  :: IORef a -> a -> IO () -- 参照への書込み
```

次に、IORef を利用したプログラムの例をあげる。

```
1 import Data.IORef
2
3 foo r = readIORef r >>= \ i ->
4         if i<=0 then return ()
5         else putStrLn (show i) >>= \ _ ->
6             writeIORef r (i-1) >>= \ _ ->
7                 foo r
8
9 main = newIORef 10 >>= \ r ->
10      foo r
```

ただし、これは無理矢理 IORef を使った例であり、IORef を使わずに同じ動作をする次のようなプログラムのほうが自然である。

```
1 {- foo の自然な定義は以下の通り -}
2 -- foo i = if i<=0 then return ()
3 --         else putStrLn (show i) >>= \ _ ->
4 --             foo (i-1)
5 --
6 -- main = foo 10
```

ところで、Haskell のプログラムを、GHCi のような対話環境ではなく、ghc コマンドで実行可能ファイルにコンパイルして実行するとき、C と同じように main という名前の関数から実行が開始されることになっている。そして main 関数は、IO t という形の型 (t は任意の型) を持たなければいけないことになっている。

たとえば、標準入力から読み込んだ文字列の大文字を小文字に変換したものと小文字を大文字に変換したものを出力するプログラムは以下のようなになる。

```
1 import Data.Char  -- Data.Char モジュールを import する
2
3 -- toLower, toUpper :: Char -> Char は大文字・小文字の変換関数
4 main :: IO ()
5 main = getContents >>= (\ s -> putStr (map toLower s)
6         >>= (\ _ -> putStr (map toUpper s)))
```

ラムダ抽象 (\... ->... ) は、どんな中置演算子よりも優先順位が低いので、上記の定義は括弧を省略し、さらにレイアウトを整えて、次のように書くことが多い。

```
1 main = getContents          >>= \ s ->
2         putStr (map toLower s) >>= \ _ ->
3         putStr (map toUpper s)
```

以降のプログラムでは、このように括弧を省略する。

実は、Haskell では Monad クラスに対して、**do 記法** という糖衣構文を用意していて、この関数は次のように書くこともできる。

```
1 main = do { s <- getContents;
2           putStr (map toLower s);
3           putStr (map toUpper s) }
```

do 式も第 A 章で紹介したレイアウトルールの対象であり、適切にレイアウトすればブレースやセミコロンを省略することができる。

```
1 main = do s <- getContents
2         putStr (map toLower s)
3         putStr (map toUpper s)
```

do 式は次のルールで翻訳される。(下線部に翻訳規則を再帰的に適用していく。)

$$\begin{aligned} \text{do } \{e\} &\Rightarrow e \\ \text{do } \{e; \textit{stmts}\} &\Rightarrow e \gg= \_ \rightarrow \underline{\text{do } \{\textit{stmts}\}} \\ \text{do } \{x \leftarrow e; \textit{stmts}\} &\Rightarrow e \gg= \ x \rightarrow \underline{\text{do } \{\textit{stmts}\}} \\ \text{do } \{\textit{let decls}; \textit{stmts}\} &\Rightarrow \textit{let decls in } \underline{\text{do } \{\textit{stmts}\}} \end{aligned}$$

Q 4.4.1 つぎのようなプログラムを上記の IO に関する関数を用いて定義せよ。

1. 一行だけ行を読み込んで、それをオウム返りするプログラム
2. 一行だけ行を読み込んで、それを 2 回オウム返りするプログラム

なお、for 文などの繰返しのための構文はないので、繰返しが必要なときは再帰関数を定義する必要がある。

---

---

---

---

---

---

---

---

問 4.4.2 IO モナドを利用して次のようなプログラムを作成せよ。(次節で紹介している関数を使用するかもしれない。)

1. 入力文字列を行毎に大文字と小文字に変換して出力する。(例えば、「Hello, World」が「hello, HELLO, world, WORLD」になる。)
2. 入力文字列中の数字 ('0' ~ '9') の出現回数をカウントして出力する。
3. 入力文字列中の '@' が出現する最初の 10 行だけを出力する。

## 4.5 続・有用なリスト処理関数

モナドとは直接関係ないが、以前紹介した以外に文字列処理プログラムで有用と思われるリスト処理関数を、以下でいくつか紹介する。これらの関数は Prelude (標準ライブラリ) に定義済みである (sort を除く)。

```

1  -- lines は文字列を改行文字のところで分割して、文字列のリストにする。
2  -- 結果の文字列には改行文字は含まれない。
3  lines      :: String  -> [String]
4
5  -- words は文字列を空白文字で分割して、文字列のリストにする。
6  words     :: String  -> [String]
7
8  -- unlines は lines の逆操作である。
9  -- 各文字列の末尾に改行文字を追加し、一つに結合する。
10 unlines   :: [String] -> String
11
12 -- unwords は words の逆操作である。
13 -- 各文字列を空白で区切って結合する。
14 unwords   :: [String] -> String
15
16 -- take n xs は xs の長さ n の先頭部分を返す。
17 -- n > length xs のときは xs 自身を返す。
18 take     :: Int -> [a] -> [a]
19
20 -- drop n xs は xs の最初の n 個の要素を除いた末尾部分を返す。
21 -- n > length xs のときは [] を返す
22 drop     :: Int -> [a] -> [a]
23
24 -- takeWhile は述語 p とリスト xs を受け取り、p を満たす要素
25 -- だけからなる xs の最も長い先頭部分（空の場合もある）を返す。
26 --
27 -- > takeWhile (< 3) [1,2,3,4,1,2,3,4] == [1,2]
28 -- > takeWhile (< 9) [1,2,3] == [1,2,3]
29 -- > takeWhile (< 0) [1,2,3] == []
30 takeWhile :: (a -> Bool) -> [a] -> [a]
31
32 -- dropWhile p xs は、takeWhile p xs の残りの末尾部分を返す。
33 --
34 -- > dropWhile (< 3) [1,2,3,4,5,1,2,3] == [3,4,5,1,2,3]
35 -- > dropWhile (< 9) [1,2,3] == []
36 -- > dropWhile (< 0) [1,2,3] == [1,2,3]
37 dropWhile :: (a -> Bool) -> [a] -> [a]
38
39 -- any は述語とリストを受け取り、リストのなかのどれかの要素が述語
40 -- を満たすかどうか判定する。
41 any      :: (a -> Bool) -> [a] -> Bool
42
43 -- all は述語とリストを受け取り、リストのなかのすべての要素が述語
44 -- を満たすかどうか判定する。
45 all     :: (a -> Bool) -> [a] -> Bool
46
47 -- リスト（空ではいけない）の先頭要素を取り出す。
48 head    :: [a] -> a
49
50 -- （非空かつ有限な）リストの最後の要素を取り出す。
51 last    :: [a] -> a

```

```

52
53 -- リスト (空ではいけない) の先頭要素を除いた末尾部分を取り出す。
54 tail           :: [a] -> [a]
55
56 -- リストの要素として含まれているか否かを判定する。
57 -- 通常 x 'elem' xs のように中置記法で用いる。
58 elem           :: (Eq a) => a -> [a] -> Bool
59
60 -- import Data.List が必要
61 -- sort 関数は安定なソートアルゴリズムの実装である。
62 -- 比較関数を引数にとる sortBy 関数もある
63 sort           :: (Ord a) => [a] -> [a]
64
65 -- sequence 関数はリスト中のアクションを順に実行する。
66 sequence       :: Monad m => [m a] -> m [a]
67
68 mapM           :: Monad m => (a -> m b) -> [a] -> m [b]
69 mapM f as      = sequence (map f as)

```

#### Q 4.5.1 次の式の値は何か?

1. drop 2 [1,4,5]
2. takeWhile (<0) [-1,-3,2,-2,7]
3. any (>0) [-1,-3,-5]
4. 3 'elem' [2,5,3,1]

## 4.6 さらに詳しく知りたい人のために...

文献 [1] は、モナドとリストの内包表記の関係について解説している。

### この章の参考文献

- [1] Philip Wadler 「Comprehending Monads」  
ACM Conference on Lisp and Functional Programming, Nice (France), 1990 年 6 月